

QUATRIEME PARTIE

INTEGRALES

EQUATIONS DIFFERENTIELLES

I - Calcul d'intégrales

1. Intégrales définies, calcul de primitives

a) Syntaxe

Int (fonction, variable)

Il s'agit d'une **forme inerte d'intégration**, sa seule utilité est d'afficher l'intégrale. Value entraîne le calcul de l'intégrale.

int (expression, variable)

Cette instruction **calcule une primitive** si cela est possible, en cas d'impossibilité, on retourne à la forme inerte.

int (expression, variable = borne inf..borne sup)

calcule l'intégrale définie entre les deux bornes dans la mesure où elle est calculable.

Exemple :

> int((1-t)/((1+t+t^2)^2),t);

$$\frac{1}{3} \frac{3t+3}{1+t+t^2} + \frac{2}{3} \sqrt{3} \arctan\left(\frac{1}{3}(2t+1)\sqrt{3}\right)$$

> int(sin(x)/x,x=0..infinity);

$$\frac{1}{2} \pi$$

b) Intégrales de fonctions ayant des pôles dans le domaine d'intégration

On calcule l'intégrale dans la mesure où cela est possible ou on retourne la valeur \pm infini.

> f:=1/sqrt(1-x**2);

$$f := \frac{1}{\sqrt{1-x^2}}$$

> **int(f,x=-1..1);**

π

Bien que la fonction ne soit pas définie aux bornes 1 et -1, l'intégrale est définie.

Essayer de calculer les intégrales ci-dessous :

> **int(f,x);**

$\arcsin(x)$

> **int(1/(x-1),x=0..infinity);**

∞

> **int(1/(x-1),x=0..1);**

$-\infty$

c) Intégrales faisant appel à des fonctions spéciales

Calculer les intégrales ci-dessous :

f:= 1/sqrt((1 - x²)(5 + x²))

int(f, x = 1/2..1)

int(exp (-x²/2), x = 0..infinity)

int(exp(x)*Dirac(x), x = -2..2)

etc...

Pas de problème parce que pratiquement toutes les fonctions spéciales sont connues par Maple.

d) Intégration numérique

Maple utilise deux méthodes. Ou bien on évalue l'expression symbolique après avoir intégré symboliquement ou bien on utilise une méthode par trapèzes ou une méthode de Simpson.

evalf(int(expression, x = borne inf..borne sup))

comme l'instruction est exécutée de l'intérieur vers l'extérieur, c'est à dire par rapport à la parenthèse la plus interne, on tentera donc nécessairement l'intégration formelle.

evalf(Int(expr, x = borne inf..borne sup))

On gagne beaucoup de temps parce qu'on va directement vers l'intégration numérique puisque la parenthèse interne comporte une intégration inerte.

Lorsque la fonction comporte un ou plusieurs pôles on a un message d'erreurs.

2. Changement de variables et intégration par parties

Lorsque Maple ne possède pas les algorithmes pour mener l'intégration au bout, on peut lui imposer la méthode à suivre. On utilise pour cela le paquetage "Student".

a) Intégration par parties

$$I = \int_a^b f(t) dt \quad \text{et } f(t) = u(t) v'(t)$$

On peut intégrer par parties à condition d'indiquer à Maple quelle est la partie de la fonction utilisée comme $u(t)$.

```
with(Student)
intparts(Int(f(t), t), u(t))
```

b) Changements de variables

```
with(Student)
changevar(var = fonct, nom, var)
```

L'intégrale doit être définie précédemment de façon inerte de préférence.

```
nom = Int(expression, varinit)
```

Exemple:

```
> with(student):
```

```
> F:=Int((sin(x)*sin(2*x))/(1+cos(x)**4+sin(x)**4),x);
```

$$F := \int \frac{\sin(x)\sin(2x)}{1 + \cos(x)^4 + \sin(x)^4} dx$$

```
> G:=changevar(u=sin(x),F,u);
```

$$G := \int \frac{u \sin(2\arcsin(u))}{(1+(1-u^2)^2 + u^4)\sqrt{1-u^2}} du$$

```
> expand(G);
```

$$\frac{1}{12}\sqrt{3}\ln(-u^2 + \sqrt{3}u - 1) + \frac{1}{2}\arctan(2u - \sqrt{3}) - \frac{1}{12}\sqrt{3}\ln(u^2 + \sqrt{3}u + 1) + \frac{1}{2}\arctan(2u + \sqrt{3})$$

On peut également faire des changements de variables plus complexes du type

$$f(x) = g(u)$$

II - Equations différentielles

1. Syntaxe

a) Syntaxe de la résolution d'équations différentielles

```
dsolve(eqdiff, solution, options)
dsolve({ensemble d'eqdiff}, {solutions}, options)
```

b) Syntaxe de l'écriture des équations différentielles

```
nom:= diff(fonc(var), var$ordre) = express(fonc)
```

Si l'ordre est de 1, \$ordre peut être omis.

Dans le cas des systèmes d'équations différentielles, il est conseillé de donner un nom aux équations différentielles afin d'alléger l'écriture. On peut également donner un nom aux différentes dérivées. Enfin, l'équation différentielle peut être donnée sous forme générale.

```
nom:= expression(fonc, dérivées, var) = 0
```

c) Options

explicit = true	donne la solution explicite autant que possible
méthod = laplace	utilisation de la transformée de Laplace
type = series	sortie des solutions sous forme de série
type = numeric	résolution numérique

2. Problème des conditions initiales

a) Résolution en fonction des constantes d'intégration

Lorsqu'on ne spécifie aucune condition initiale, la solution est retournée en fonction de une ou de plusieurs constantes d'intégration. Celles-ci ont toujours pour nom :

$$_C_1, _C_2, _C_3, \dots \text{ etc}$$

b) Introduction des conditions initiales dans l'équation

La méthode de résolution précédente n'est pas la meilleure puisqu'il faut refaire une résolution algébrique pour déterminer les valeurs des $_C_i$. On peut introduire les conditions initiales directement dans le dsolve.

```
dsolve({eq.diff, cond.init}, solutions, options)
```

Exemple :

Cinétique de trois réactions consécutives : $A \xrightarrow{k_1} B \xrightarrow{k_2} C$

> ed1:=diff(A(t),t)=-k1*A(t):

> ed2:=diff(B(t),t)=k1*A(t)-k2*B(t):

> ed3:=diff(C(t),t)=k2*B(t):

> cond1:=A(0)=a:

> cond2:=B(0)=0:

> cond3:=C(0)=0:

> sysdiff:={ed1,ed2,ed3}:

> sysdiff:=sysdiff union {cond1,cond2,cond3}:

> dsolve(sysdiff,{A(t),B(t),C(t)});

$$\left\{ C(t) = a + \frac{a k_2 e^{(-k_1 t)}}{-k_2 + k_1} - \frac{k_1 a e^{(-k_2 t)}}{-k_2 + k_1}, B(t) = -\frac{a k_1 e^{(-k_1 t)}}{-k_2 + k_1} + \frac{k_1 a e^{(-k_2 t)}}{-k_2 + k_1}, A(t) = a e^{(-k_1 t)} \right\}$$

3. Méthodes numériques**a) Syntaxe et options**

dsolve({eqdiff}, {solutions}, type = numeric)

Il faut choisir entre deux méthodes

method = rkf45	Runge-Kutta (par défaut)
method = dverk72	
method = gear	méthode de Gear
method = mgear	Méthode de Gear multi-pas
method = isode	Méthode spéciale pour les équations raides

b) Sortie

La sortie d'une résolution numérique est toujours une procédure. On utilise ensuite cette procédure pour créer une liste de valeurs numériques x, y, x, y qui représentent la solution de l'équation diff.

Exemples :

> `ed:=diff(y(x),x$2)=-sin(y(x));` # équation du pendule

$$ed := \frac{d^2}{dx^2} y(x) = -\sin(y(x))$$

> `sol:=dsolve({ed,y(0)=0,D(y)(0)=0.5},y(x),numeric);`

`sol:=proc(rkf45_x)... end`

> `sol(1);sol(2);` # donne les solutions point par point


$$\left[x = 1, y(x) = .4245341926928455, \frac{d}{dx} y(x) = 0.2737235288022365 \right]$$

$$\left[x = 2, y(x) = .4663684126590473, \frac{d}{dx} y(x) = -.1908250824271950 \right]$$

> `f(t):=t→subs(sol(t),y(x));` transforme la procédure en fonction

`f(t):= t → subs(sol(t), y(x))`

> `plot(f(t),0..10,title='Equation du pendule sans approximations');`



4. Tracé direct des courbes correspondant à une équation ou un système différentiel

On peut tracer directement les résultats d'une résolution numérique d'une équation différentielle sans avoir à calculer la courbe point par point.

a) Utilisation de odeplot

```
with(plots)                chargement du paquetage
nom:= dsolve({eqdiff, condinit}, fonc(var), numeric)
nom:= proc(rkf45_x)..end
odeplot(nom, [var, fonc(var)], début..fin)
```

La fonction est tracée entre début et fin **sans en connaître l'équation explicite**.

b) Utilisation de DEplot

Le paquetage DEtools est plus puissant que plots pour le tracé de solutions d'équations différentielles.

```
with(DEtools)              chargement du paquetage
nom:= équation différentielles sans condition initiale
```

```
DEplot(nom, [var, fonc(var)], debut..fin, {condinit}, options)
```

var et fonc(var)	doivent correspondre à l'équation différentielle
{condinit}	ensemble de listes
[varinit, valfunc]	structure des listes contenues dans {condinit}: varinit est la valeur initiale de la variable, valfunc la valeur correspondante de la fonction

Options de DEplot

stepsize = h	pas d'intégration
y = c..d	calibre de l'axe vertical (obligatoire)
scene = [x, y]	indique que c'est y(x) qui doit être tracé
type = arrows	permet de tracer un champ de vecteurs représentant la solution générale de l'équation différentielle
type = line	trace une famille de courbes

c) Utilisation de DEplot pour un système différentiel

```
nomsys:= [système différentiel donné sous forme de liste]
```

Si l'option scene n'est pas spécifiée, Maple retourne un tracé 3D de fonc1, fonc2 en fonction de var.

Si on spécifie :

scene = [var, fonc1] ou scene = [var, fonc2]

on trace respectivement fonc1 ou fonc2 en fonction de la variable var.


Exemples :

> with(DEtools):

> sysdiff:=[diff(x(t),t)=x(t)/12-y(t),diff(y(t),t)=x(t)-y(t)/12];

$$\text{sysdiff} := \left[\frac{d}{dt} x(t) = \frac{1}{12} x(t) - y(t), \frac{d}{dt} y(t) = x(t) - \frac{1}{12} y(t) \right]$$

> DEplot3d(sysdiff,[x,y],t=0..4*Pi,[0,1,1],stepsize=0.1);



DEplot(sysdiff,[x,y],t=0..4*Pi,[0,1,1],stepsize=0.1,scene=[x,y],arrows=line);

